
Plot()
 {
  Curve("IntCal20","IntCal20.14c");
  Curve("SHCal20","SHCal20.14c");
  Mix_Curve("Mixed","IntCal20","SHCal20",U(0,100));
  Sequence("Iruhito Sequence")
  {
   Boundary("MF Start");

   {
    Sequence("Ribera 1–2 Sequence")
    {
     R_Date("AA75520, U4.2, Feature 8, earliest occupation surface",2493,35);
     R_Date("AA75519, U4.2, Feature 7, occupation surface",2458,34);
    };
   };
   Boundary("MF End");
   Interval("Gap between Ribera 2 and and 3");
   Boundary("ILF Start");
   KDE_Plot("Ribera 3–4")
   {
    Sequence()
    {
     R_Date("D-AMS005618, U4.17, 417.24, earliest occupation surface above sterile in 4.17",2171,25);
     R_Date("D-AMS005619, U4.15, 415.20, thin midden associated with Ribera 4 limestone wall",2135,25);
    };
   };
   KDE_Plot("Ribera 5")
   {
    R_Date("D-AMS005617, U4.12, 412.9, deposition associated with Ribera 5 adobe wall",2075,27);
    R_Date("D-AMS005621, U4.15, 415.9, fill over occupation surface associated with Ribera 5",2049,29);
    R_Date("D-AMS005616, U4.15, 415.15, hearth feature",2043,29);
    R_Date("AA75518, U4.2, Feature 3, final Ribera occupation surface, fragmentary adobe wall",2004,40);
   };
   Boundary("ILF End");
   Interval("Gap between ILF and Post-LF");
   Boundary("Start Post LF ");
   KDE_Plot("Post-LF")
   {
    Sequence("Sector 5 Sequence")
    {
     R_Date("AA110492, U5.1, Level 13, earliest occupation in Sector 5 Monticulo",1372,27);
     Date("First Tiwanaku redwares");
     R_Date("AA75516, U5.1, Feature 9, midden between levels 8 and 9",1278,33);
     R_Date("AA75517, U5.1, Feature 8, floor between levels 7 and 8",1231,33);
    };
   };
   Boundary("End Post-LF");
  };
 };
